i21edu/Mod/ВыводПример0.odc

МОДУЛЬ i21eduВыводПример0;    (* см. замечания внизу после текста модуля *)
	ПОДКЛЮЧИТЬ  Вывод := i21eduВывод, Math;
	
	ПЕРЕМЕННЫЕ  ц: ЦЕЛАЯ;  л: ЛИТЕРА;  в: ВЕЩЕСТВЕННАЯ;
		цепь: МАССИВ 100 ИЗ ЛИТЕР; (* подобные массивы служат для хранения цепочек литер; в данный массив поместятся цепочки длиной не более 99 литер *)
		
НАЧАЛО
	Вывод.НовСтрока;	(* перевод дальнейшей печати на новую строку *)
	Вывод.Цепочка( '***********' );	(* чтобы легче найти начало печати в журнале *)
	Вывод.НовСтрока;	
	
	(* вывод целых: *)
	Вывод.Цел( 5 );	(* целая константа *)
	ц := 17;
	Вывод.Цел( ц ); 	(* значение целой переменной *)
	Вывод.Цел( ц - 25 ); 	(* выражение, дающее целый результат *)
	Вывод.Цел( ц DIV 2 ); 	(* деление DIV даёт целое, остаток отбрасывается *)
	Вывод.Цел( ц MOD 2 ); 	(* остаток от деления на 2 *)
	Вывод.НовСтрока;
	
	(* вывод литер: *)
	Вывод.Литера( "Ю" );	(* литера-константа *)
	л := ";";
	Вывод.Литера( л );	(* значение литерной переменной *)
	Вывод.НовСтрока;
	
	(* вывод литерных цепочек: *)
	Вывод.Цепочка( "два слова" );	(* цепочка-константа *)
	Вывод.Цепочка( '--можно и в одинарных кавычках' );
	цепь := "; ABC";	(* заряжаем в массив цепочку *)
	Вывод.Цепочка( цепь );	(* цепочка из массива *)
	Вывод.НовСтрока;
	Вывод.Цепочка( "два слова"+цепь );	(* цепочки можно складывать (конкатенация) *)
	Вывод.НовСтрока;
	
	(* вывод вещественных чисел: *)
	Вывод.Вещ( 3.14 );	(* вещественная константа *)
	в := 100;
	Вывод.Вещ( в );	(* значение вещественной переменной *)
	Вывод.Вещ( - в * 0.01 );	(* вещественное выражение *)
	Вывод.Вещ( в / 3 );	(* деление дробью всегда даёт вещественный 
		результат *)
	Вывод.Вещ( Math.Sqrt( 100 ) );	(* значение функции квадратный корень от 100 *)
	Вывод.НовСтрока;

КОНЕЦ i21eduВыводПример0.

Перевод ключевых слов на английский: Программирование, Перевести в English

Англоязычный аналог этого примера со взрослым модулем StdLog: i21eduStdLogПример0.

Скомпилировать и выполнить: Ctrl+F9.

Результаты печатаются в Рабочий журнал, в его конец. 
Если журнал не виден, то нужно выполнить команду меню 1-Основное, Открыть Рабочий журнал (та же команда есть в штатном меню Инфо; в англ. варианте Info, Open Log). Можно изменить размеры окна журнала по желанию.

Обычно лучше сохранять в журнале историю своих действий. Но если всё-таки потребуется его очистить, то нужно пойти в штатное меню Инфо и вызвать там команду Очистить журнал.

Напоминание:

 Шпаргалку по средствам любого модуля можно получить, выделив имя модуля в любом текстовом документе (например, в Рабочем журнале/Log) и нажав Ctrl+D. Во вспомогательном окошке откроется список всех объектов, предоставляемых модулем (процедур, переменных, констант и типов)  но без подробных объяснений.

 Полную документацию можно почитать, нажав Ctrl+Shift+D вместо Ctrl+D после выделения имени модуля (или через меню: Инфо, Документация).
Поскольку любые переводы чреваты ошибками, имеет смысл помнить о том, что последней инстанцией является английская версия документации; она открывается командой меню Инфо, Documentation.

 Указанные выше комбинации быстрого вызова (Ctrl+Shift+D и т.п.) справедливы для школьных комплектов Блэкбокса от проекта Информатика-21. В других комплектах они могут быть другими.